
บทที่   Â
ฟงกชัน (Function)

จดุประสงคเชิงพฤติกรรม
1. เขาใจพื้นฐานการฟงกชันใชงาน
2. ใชอางองิเมื่อมีความตองการ  แตนึกรูปแบบฟงกชันไมออก
3. ใชอางองิเมื่อตองการหาฟงก็ชัน ไปแกปญหาในการเขียนโปรแกรม

หวัขอในบทเรียน
ฟงกชันทั้งหมด 194 ฟงกชัน
จาก ฟงกชัน AADD () จนถึง ฟงกชัน YEAR()
ตวัอยางโปรแกรมเกี่ยวกับฟงกชัน  227 ตัวอยาง

หมายเหตุ
มฟีงกชันใชงานมากมายที่ Clipper มใีหบรกิาร มีฟงกชันหลายตัวที่มีหนาที่คลายกันทั้งในฟงกชัน หรือค ํา

สัง่เอง กเ็พือ่สรางทางเลือกใหกับผูใชในการเขียนโปรแกรม และหลายคนมีพื้นฐานมาจากภาษาอื่น ก็สามารถเลือกใช
ฟงกชนัทีต่นคุนเคยไดโดยงาย บทนี้ตองการเปนแหลงอางอิง เพราะมีนักเรียน หรือนักพัฒนานอยคน ที่จะจํ าฟงกชัน
ทัง้หมดไดครบ หรือจ ํารูปแบบการใชไดทั้งหมด .. แหลงอางอิงจึงมีความสํ าคัญ

บทที่ 3   ฟงกชัน (Function)
ฟงกชนั คอืการประมวลผลกับคาที่สงใหฟงกชันดํ าเนินการ  แลวคืนคาที่ผานการประมวลผลกลับมา   หรือ

การหาคาโดยใชฟงกชัน    นั่นหมายความวา ฟงกชันมีหนาที่คืนคาใหกับโปรแกรม  แลวโปรแกรมสามารถน ําคาไปใช
ไดตามตองการ  แตจะคือคามาในรูปแบบใด ขึ้นอยูกับหนาที่ของฟงกชันนั้น ๆ

&  3.1 AADD()  10

! รูปแบบ
AADD(<ชดุอาเรยเปาหมาย>,<นิพจนที่เพิ่มเขาไป>) --> คาของนิพจน
P หนาที่
ฟงกชนันี้จะทํ าหนาที่ เพิ่มสมาชิกอีก 1 ตัวเขาไปตอทายอาเรยชุดเดิมโดยจะเพิ่มคาเขาไปตออาเรยตัวสุดทาย แลวสงคา
คนื ซึง่เปนคาของนิพจนที่เพิ่มเขาไป
: ตัวอยางที่ 3.1
        AR := {}                // อาเรยวางเปลา
        AADD(AR,'BAT')          // {'BAT'}
        AADD(AR,'RAT')          // {'BAT','RAT'}
        AADD(AR,2500)           // {'BAT','RAT',2500}
        FOR I = 1 TO 3
           ?? AR[I]             // BATRAT      2500
        NEXT
: ตัวอยางที่ 3.2
        AR := {}
        TEST := 'ABC'
        AADD(AR,'A')
        AADD(AR,5)
        ? AADD(AR,'XYZ')        // XYZ
        AADD(AR,{'B',TEST,23})
        ? AR[1]                 // A
        ? AR[2]                 //          5
        ? AR[3]                 // XYZ
        ? AR[4,1]               // B
        ? AR[4,2]               // ABC
        ? AR[4,3]               //         23
        ? '====='               // =====

10  นิพนธ กิตติปภัสสร และ อนันต  อุตตะมะ,  อางแลวในเชิงอรรถที่ ( 1 ),  หนา 135

&  3.2 ABS()
! รูปแบบ
ABS(<นพิจนจ ํานวนเลข>) --> จํ านวนเลขบวกหรือศูนย
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ แปลงตัวแปรตัวเลขเปนคาสัมบูรณ
: ตัวอยางที่ 3.3
        TEST1 := 10
        TEST2 := 12
        ? (TEST1 - TEST2)       //         -2
        ? ABS(TEST1 - TEST2)    //          2
        ? ABS(TEST1 - 100)      //         90
        ? ABS(3 - 7)            //          4
        ? ABS(0)                //          0
        ? ABS(5)                //          5
        ? ABS(22 / 7)           //          3.14
        ? ABS(2.555 - 3.4444)   //          0.8894

&  3.3 ACHOICE()
! รูปแบบ
ACHOICE(<มมุบน>, <มุมซาย>, <มุมลาง>, <มุมขวา>,
        <ชดุอาเรยแสดงเมน>ู) --> จํ านวนเลขของตัวเลือกที่ถูกเลือก
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งใหนํ า ARRAY มาเปนเมนูแบบ PULLDOWN MENU
: ตัวอยางที่ 3.4
        CLS
        DO WHILE .T.
           OPT := ACHOICE(5,10,7,30,{'ADD','DELETE';
                  ,'CHANGE','PRINT','QUIT'})
           OPTPROGRAM := {'ADDPRO','DELPRO','CHGPRO','PRTPRO'}
           DO CASE
              CASE OPT = 1
                     DO ADDPRO
              CASE OPT = 2

                     DO DELPRO
              CASE OPT = 3
                     DO CHGPRO
              CASE OPT = 4
                     DO PRTPRO
              CASE OPT = 5
                     EXIT
           ENDCASE
        ENDDO
: ตัวอยางที่ 3.5
        CLS
        OPTPROGRAM := {'ADDPRO','DELPRO','CHGPRO','PRTPRO'}
        DO WHILE .T.
           OPT := ACHOICE(5,10,7,30,{'ADD','DELETE';
                  ,'CHANGE','PRINT','QUIT'})
           IF OPT = 5 .OR. LASTKEY() = 27
              EXIT
           ENDIF
           ? (OPTPROGRAM[OPT])
           DO CASE
              CASE OPT = 1;   DO ADDPRO
              CASE OPT = 2;   DO DELPRO
              CASE OPT = 3;   DO CHGPRO
              CASE OPT = 4;   DO PRTPRO
           ENDCASE
        ENDDO
: ตัวอยางที่ 3.6
        CLS
        OPTMENU    := {'ADD','DELETE','CHANGE','PRINT','QUIT'}
        OPTPROGRAM := {'ADDPRO','DELPRO','CHGPRO','PRTPRO'}
        DO WHILE .T.
           OPT := ACHOICE(5,10,7,30,OPTMENU)
           DO CASE

              CASE OPT = 5 .OR. LASTKEY() =27 ; EXIT
              CASE OPT = 1;   DO ADDPRO
              CASE OPT = 2;   DO DELPRO
              CASE OPT = 3;   DO CHGPRO
              CASE OPT = 4;   DO PRTPRO
           ENDCASE
        ENDDO

&  3.4 ACLONE()
! รูปแบบ
ACLONE(<ชดุอาเรยตนแบบ>) --> ชุดอาเรยชุดใหม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คัดลอกอาเรยทั้งชุดไปสรางเปนชุดใหม
: ตัวอยางที่ 3.7
        OLD := {'ABC',12,2.5}
        NEW := OLD                      // 2 ตวัแปร แตขอมูลชุดเดียว
        OLD[1] := 'DEF'
        ? OLD[1], OLD[2], OLD[3]        // DEF  12   2.5
        ? NEW[1], NEW[2], NEW[3]        // DEF  12   2.5
: ตัวอยางที่ 3.8
        OLD := {'ABC',12,2.5}
        NEW := ACLONE(OLD)              // เก็บขอมูลไว 2 ชุด
        OLD[1] := 'DEF'
        ? OLD[1], OLD[2], OLD[3]        // DEF  12   2.5
        ? NEW[1], NEW[2], NEW[3]        // ABC  12   2.5

&  3.5 ACOPY()
! รูปแบบ
ACOPY(<ชดุอาเรยตนแบบ>, <ชุดอาเรยเปาหมาย>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>],
      [<ต ําแหนงปลายทาง>]) --> ชุดอาเรยปลาย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คัดลอกอาเรยไปทับอาเรยชุดใหม
: ตัวอยางที่ 3.9

        OLD := {'A','B','C'}
        NEW := {1,2,3}
        ACOPY(OLD,NEW,1,1,1)    // OLD ตวัแรกตัวเดียวทับ NEW ตัวเดียว
        ? NEW[1],NEW[2],NEW[3]  // A   2   3
        ACOPY(OLD,NEW,1,1,2)    // OLD ตวัแรกทับ NEW 2 ชุด ไปต ําแหนงที่ 2
        ? NEW[1],NEW[2],NEW[3]  // A   A   3
        ACOPY(OLD,NEW,1,2,1)    // OLD จากตัวแรกไป 2 ตัว ไปต ําแหนงที่ 1
        ? NEW[1],NEW[2],NEW[3]  // A   B   3
        ACOPY(OLD,NEW,2,2,1)   // OLD จากตัวที ่2 ไป 2 ตัว ไปต ําแหนงที่ 1
        ? NEW[1],NEW[2],NEW[3]  // B   C   3
        ACOPY(OLD,NEW,3,1,1)    // OLD จากตัวที ่3 ไป 1 ตัว ไปต ําแหนงที่ 1
        ? NEW[1],NEW[2],NEW[3] // C   C   3
        ACOPY(OLD,NEW,1,1,3)    // OLD จากตัวที ่1 ไปต ําแหนงที่ 3
        ? NEW[1],NEW[2],NEW[3]  // C   C   A

&  3.6 ADEL()
! รูปแบบ
ADEL(<ชดุอาเรยเปาหมาย>, <ตํ าแหนงที่>) --> ชุดอาเรยปลายทาง
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ลบอาเรยบางตัวออกจากชุดอาเรยที่มีอยู
: ตัวอยางที่ 3.10
        OLD := {'A','B','C'}
        NEW := {1,2,3}
        ACOPY(OLD,NEW,1,1,3)
        ? NEW[1],NEW[2],NEW[3]                  //  1  2  A
        ADEL(OLD,2)
        ADEL(NEW,3)
        ? ; FOR I = 1 TO 3 ; ?? OLD[I] ; NEXT   //  A  C  NIL
        ? ; FOR I = 1 TO 3 ; ?? NEW[I] ; NEXT   //  1  2  NIL

&  3.7 ADIR()
! รูปแบบ
ADIR([<ชือ่แฟมที่ตองการอานเขามา>],
     [<ชดุอาเรยที่รับชื่อแฟม>],     [<ชุดอาเรยที่รับขนาด>],
     [<ชดุอาเรยที่รับวันที>่],     [<ชุดอาเรยที่รับเวลา>],
     [<ชดุอาเรยทีร่ับลักษณะเฉพาะ>]) --> จํ านวนแฟมที่รับเขามา
P หนาที่
ฟงกชนันี้จะท ําหนาที่ น ําขอมูลแฟมเชน ชื่อแฟม หรือขนาด เก็บลงอาเรย
: ตัวอยางที่ 3.11
        LOCAL FPRG[20]               // จองเนื้อที่เปนอาเรยให FPRG
        LOCAL FPRGTEST := ARRAY(20)  // จองเนื้อที่เปนอาเรยอีกแบบหนึ่ง
        ADIR("*.PRG",FPRG)
        FOR I = 1 TO 3               // X.PRG
            ? FPRG[I]                // Y.PRG
        NEXT                         // Z.PRG
: ตัวอยางที่ 3.12
        LOCAL FINDDIR := "C:\WINDOWS\*.EXE"
        LOCAL FDIRS[ADIR(FINDDIR)]   // ท ําใหจองเนื้อที่ใหอาเรยไดถูกตอง
        ADIR(FINDDIR,FDIRS)
        AEVAL(FDIRS,{|ELEMENT| QOUT(ELEMENT)}) // QOUT ใหพิมพอาเรยได
: ตัวอยางที่ 3.13
        LOCAL FINDDIR := "C:\WINDOWS\*.EXE"
        LOCAL FDIRS[ADIR(FINDDIR)] // ท ําใหจองเนื้อที่ใหอาเรยไดถูกตอง
        LOCAL FSIZES [ADIR(FINDDIR)]
        LOCAL FDATES [ADIR(FINDDIR)]
        LOCAL FTIMES [ADIR(FINDDIR)]
        LOCAL FATTRIBUTES [ADIR(FINDDIR)]
        ADIR(FINDDIR,FDIRS,FSIZES,FDATES,FTIMES,FATTRIBUTES)
        AEVAL(FDIRS ,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FSIZES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FDATES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FTIMES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)
        AEVAL(FATTRIBUTES,{|ELEMENT| QOUT(ELEMENT)}) ; INKEY(0)

: ตัวอยางที่ 3.14
        LOCAL FINDDIR
        ACCEPT TO FINDDIR
        FDIRS  = ARRAY(ADIR(FINDDIR))
        FSIZES  = ARRAY(ADIR(FINDDIR))
        FDATES  = ARRAY(ADIR(FINDDIR))
        FTIMES  = ARRAY(ADIR(FINDDIR))
        FATTRIBUTES = ARRAY(ADIR(FINDDIR))
        ADIR(FINDDIR,FDIRS,FSIZES,FDATES,FTIMES,FATTRIBUTES)
        FOR I = 1 TO ADIR(FINDDIR)
           ? FDIRS[I],FSIZES[I],FDATES[I],FTIMES[I],FATTRIBUTES[I]
        NEXT

&  3.8 AEVAL()
! รูปแบบ
AEVAL(<ชดุอาเรย>, <นิพจนที่น ําอาเรยไปใช>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>]) --> คาจากนิพจน
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งประมวลผลกับอาเรยดวยชุดคํ าสั่งแบบบล็อก(BLOCK)
: ตัวอยางที่ 3.15
        LOCAL ELE    := "{|ELEMENT| QOUT(ELEMENT)}"
        LOCAL ELEMIN :="{|ELEMENT| IF(ELEMENT<MIN,MIN:=ELEMENT,) }"
        LOCAL ELECNT :="{|ELEMENT| CNT:=CNT+1}"
        // LOCAL ELECNT:="{|ENIL|IF(ENIL<>NIL,CNT:=CNT+1)}"เฉพาะที ่!= NIL
        LOCAL ELESUM :="{|ELEMENT| SUM := SUM + ELEMENT}"
        MAX = 0
        MIN = 9999
        CNT = 0
        SUM = 0
        X := {40,12,8,5,21}
        Y := ACLONE(X)
        AEVAL( X , {|ELEMENT| QOUT(ELEMENT)})  ; INKEY(0)
        AEVAL( X , &ELE)                       ; INKEY(0)
        AEVAL( X , {|ELEMENT| IF(ELEMENT > MAX ,MAX := ELEMENT,)})

        AEVAL( X , &ELEMIN)
        AEVAL( X , &ELECNT)
        AEVAL( X , &ELESUM)
        ? MAX , MIN , CNT , SUM/CNT

&  3.9 AFIELDS()
! รูปแบบ
AFIELDS([<ชดุอาเรยเก็บชื่อฟลด>], [<ชุดอาเรยเก็บแบบ>],
        [<ชดุอาเรยเกบ็ความกวาง>], [<ชุดอาเรยเก็บจํ านวนทศนิยม>])
        --> จ ํานวนฟลดของแฟมปจจุบัน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ใสโครงสรางแฟมปจจุบันลงไปใน ARRAY
: ตัวอยางที่ 3.16
        USE FILEA
        AFLDNAME := ARRAY(AFIELDS())
        AFLDTYPE := ARRAY(AFIELDS())
        AFLDWIDTH:= ARRAY(AFIELDS())
        AFLDDEC  := ARRAY(AFIELDS())
        AFIELDS(AFLDNAME,AFLDTYPE,AFLDWIDTH,AFLDDEC)
        FOR I = 1 TO AFIELDS()                  // FIELD1 C  10
            ?  AFLDNAME[I],AFLDTYPE[I]          // FIELD2 N  10  0
            ?? STR(AFLDWIDTH[I],10)             // FIELD3 N  10  0
            ?? IF(AFLDTYPE[I]='N',STR(AFLDDEC[I],10),'')
        NEXT

&  3.10 AFILL()
! รูปแบบ
AFILL(<ชดุอาเรยเปาหมาย>, <คาที่จะนํ าเขาชุดอาเรย>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>]) --> ชุดอาเรยเปาหมาย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ใสคาที่กํ าหนดลงไปใน ARRAY โดยรูปแบบขอมูลแตละครั้งก ําหนดไดเพียงแบบเดียว
: ตัวอยางที่ 3.17
        LOCAL  X[5]     // NIL NIL NIL NIL NIL
        LOCAL  Y[5]     // NIL NIL NIL NIL NIL
        AFILL(X,12)     // น ําเลข 12 ใสในอาเรยทุกตัว
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 12 12 12 12 12

        AFILL(X,5,2,3)    // ใสเลข 5 เริ่มที่ตํ าแหนง 2 ไป 3 ตัว
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 12  5  5  5 12
        AFILL(X,.T.,3,2)  // ใส .T. เริ่มที่ตํ าแหนง 3 ไป 2 ตัว
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 12  5 .T. .T. 12

&  3.11 AINS()
! รูปแบบ
AINS(<ชดุอาเรยเปาหมาย>, <ตํ าแหนงที่>) --> ATARGET
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แทรกคาวาง(NIL) ลงไปในอาเรย ท ําใหตัวสุดทายหายไปเสมอ
: ตัวอยางที่ 3.18
        X := {1,2,3,4,5}
        AINS(X,2)                       // แทรกเขาไปในตํ าแหนงที่ 2
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 1  NIL  2  3  4
        AINS(X,4)                       // แทรกเขาไปในตํ าแหนงที่ 4
        FOR I = 1 TO 5 ; ?X[I] ; NEXT   // 1  NIL  2  NIL  3

&  3.12 ALERT()
! รูปแบบ
ALERT(<ขอความ> [, <ชุดอาเรยตัวเลือก>]) --> คาตัวเลือก
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ การรับคาจากตามตัวเลือกในแบบไดอาลอกบล็อก (DIALOG BOX)
: ตัวอยางที่ 3.19
        X = {'A','B','C','D'}           // ตองไมเกิน 4 คา
        OPT = ALERT('TEST ALERT',X)
        ? OPT                           // ผลคือ 1 ถึง 4 เทานั้น
        HEADOPT = 'ตดัสินใจส ําหรับรายงานนี้'
        TXTOPT = {'SCREEN','PRINTER','EDIT OPTION','CANCEL'}
        OPT = ALERT(HEADOPT,TXTOPT)
        DO CASE
           CASE OPT = 1 ; ? 'พมิพรายงานทางจอภาพ'
           CASE OPT = 2 ; ? 'พมิพรายงานทางเครื่องพิมพ'

           CASE OPT = 3 ; ? 'กลับไปแกไขตัวเลือก'
           CASE OPT = 4 ; ? 'ยกเลิก และกลับเมน'ู
           OTHERWISE    ; ? 'ทานไมเลือกปุมใดเลย รูตัวรึเปลา'
        ENDCASE
: ตัวอยางที่ 3.20
        HEADOPT = 'ตดัสนิใจส ําหรับรายงานนี้ และไมอนุญาติใหกด ESC'
        TXTOPT = {'SCREEN','PRINTER','EDIT OPTION','CANCEL'}
        OPT = 4
        DO WHILE .T.
           OPT = ALERT(HEADOPT,TXTOPT)
           DO CASE
              CASE OPT = 1 ; ? 'พมิพรายงานทางจอภาพ'
              CASE OPT = 2 ; ? 'พมิพรายงานทางเครื่องพิมพ'
              CASE OPT = 3 ; ? 'กลับไปแกไขตัวเลือก'
              CASE OPT = 4 ; ? 'ยกเลิก และกลับเมน'ู; EXIT
              OTHERWISE
                ALERT('ทานไมเลือกปุมใดเลย รูตัวรึเปลา')
           ENDCASE
        ENDDO

&  3.13 ALIAS()
! รูปแบบ
ALIAS([<พืน้ทีท่ ํางานที่>]) --> สมนาม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงชื่อพื้นที่ท ํางานปจจุบัน
: ตัวอยางที่ 3.21
        SELE 1
        USE FILEA
        ? ALIAS()                  // FILEA
        SELE 2
        USE FILEB ALIAS AREANAME2
        ? ALIAS()                  // AREANAME2
        SELE 3
        USE FILEC ALIAS AREANAME3  // ชื่อ ALIAS ตองไมเกิน 10

        ? ALIAS()                  // AREANAME3
        SELE AREANAME2
        ? ALIAS()                  // AREANAME2

&  3.14 ALLTRIM()
! รูปแบบ
ALLTRIM(<ขอความ>) --> ขอความที่ผานการตัดชองวาง
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ น ําคาวางออกจากตัวแปรตัวอักษร ทั้งดานซาย และดานขวา
: ตัวอยางที่ 3.22
        X = "ABC "
        Y = " ABC"
        Z = " ABC "
        ? LEN(ALLTRIM(X))                       //          3
        ? ALLTRIM(X)+ALLTRIM(Y)                 // ABCABC
        ? LEN(LTRIM(X))                         //          4
        ? LEN(RTRIM(X))                         //          3
        ? LEN(LTRIM(Z))                         //          4
        ? LEN(RTRIM(Z))                         //          4
        ? LEN(ALLTRIM(Z))                       //          3
        ? ALLTRIM(X)+ALLTRIM(Y)+ALLTRIM(Z)  // ABCABCABC
        ? RTRIM(X)+RTRIM(Y)+RTRIM(Z)            // ABC ABC ABC
        ? LTRIM(X)+LTRIM(Y)+LTRIM(Z)            // ABC ABCABC

&  3.15 ARRAY()
! รูปแบบ
ARRAY(<จ ํานวนสมาชกิ> [, <จํ านวนสมาชิก>...]) --> ชุดอาเรย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ กํ าหนดใหมีการใชตัวแปร ARRAY
: ตัวอยางที่ 3.23
        AR := ARRAY(3)          // การก ําหนดอาเรยมิติเดียวมาใช
        AR := {NIL , NIL , NIL} // หรือก ําหนดแบบใช {}
        FOR I = 1 TO 3
            ACCEPT TO AR[I]

        NEXT
        FOR I = 1 TO 3
            ? AR[I]
        NEXT
: ตัวอยางที่ 3.24
        AR := ARRAY(2,3)        // การก ําหนดอาเรยหลายมิติมาใช
        AR := {{NIL,NIL,NIL},{NIL,NIL,NIL}}
        AR := {ARRAY(3),ARRAY(3)}
        // น ําขอมูลจากโครงสรางแฟมเก็บในอาเรยหลายมิติ
        USE FILEA
        FLDAR := ARRAY(4,AFIELDS())
        AR1 := FLDAR[1]
        AR2 := FLDAR[2]
        AR3 := FLDAR[3]
        AR4 := FLDAR[4]
        AFIELDS(AR1,AR2,AR3,AR4)
        FOR I = 1 TO AFIELDS()
            ? FLDAR[1,I],FLDAR[2,I],FLDAR[3,I],FLDAR[4,I]
        NEXT
        FOR I = 1 TO AFIELDS()
            ?
            FOR J = 1 TO 4
                ?? FLDAR[J,I]
            NEXT
        NEXT
        AEVAL(FLDAR,{|ELE|QOUT(ELE[1],ELE[2])})

&  3.16 ASC()
! รูปแบบ
ASC(<ตวัอกัษร>) --> เลขล ําดับในตารางแอสกี (ASCII)
P หนาที่
ฟงกชนันี้จะทํ าหนาที่ แปลงคาใหเปนรหัส ASCII
: ตัวอยางที่ 3.25
        ? ASC("A")              //         65
        ? ASC("A")             //         97
        ? ASC("ABCDEF")         //         97
        ? ASC("Z")              //         90

        ? ASC("Z123")           //         90
        ? ASC("Z")              //        122
        ? ASC("Z") - ASC("A")   //         25
        ? ASC("Z") - ASC("A")   //         25
        FOR I = ASC("A") TO ASC("Z")
            ? I                 //    97 - 122
        NEXT

&  3.17 ASCAN()
! รูปแบบ
ASCAN(<ชดุอาเรยเปาหมาย>, <ขอความที่ตองการหา>,
      [<เริม่ทีต่ ําแหนง>], [<จํ านวนตํ าแหนง>]) --> เลขแสดงตํ าแหนงที่พบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คนหาคาที่ตองการในอาเรย
: ตัวอยางที่ 3.26
        AR := {'LION','TIGER','STAR','COCONUT','KEYBOARD'}
        ? ASCAN(AR,'LION')              //         1
        ? AR[ASCAN(AR,'LION')]          //    LION
        ? ASCAN(AR,'LION')              //         0

&  3.18 ASIZE()
! รูปแบบ
ASIZE(<ชดุอาเรยเปาหมาย>, <เลขกํ าหนดจํ านวนสมาชิก>) --> ชุดอาเรยเปาหมาย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งกํ าหนดขนาดของชุดอาเรย ซึ่งเพิ่มหรือลดจากเดิมได
: ตัวอยางที่ 3.27
        AR  := {'A',1}
        CNT := 0
        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 2
        ASIZE(AR,4)             // A   1   NIL   NIL
        CNT := 0
        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 4
        ASIZE(AR,3)             // A   1   NIL
        CNT := 0

        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 3
        ASIZE(AR,CNT-1)         // A   1
        CNT := 0
        AEVAL(AR,{|ELEMENT| CNT:=CNT+1})  ; ? CNT    // 2
        FOR I = 1 TO CNT
            ? AR[I]
        NEXT

&  3.19 ASORT()
! รูปแบบ
ASORT(<ชดุอาเรยเปาหมาย>, [<เริ่มที่ตํ าแหนง>],
      [<จ ํานวนต ําแหนง>], [<ตรรกเลือกแบบในการจัดเรียง>]) --> ชุดอาเรยเปาหมาย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งจัดเรียงสมาชิกในอาเรย
: ตัวอยางที่ 3.28
        AR := {5,3,1,2,4}
        X  := ACLONE(AR)
        ASORT(AR)                  // จดัเรยีงสมาชิกทุกตัวจากนอยไปมาก
        ASORT(AR,,,{|X,Y| X > Y})  // จดัเรยีงสมาชิกทุกตัวจากมากไปนอย
        ASORT(AR,2,3)              // จดัเรยีงตัวที ่2 ถึง 4 จากนอยไปมาก
        AEVAL(AR,{|ELE|QOUT(ELE)}) // 5 2 3 4 1
: ตัวอยางที่ 3.29
        AR := ARRAY(4,2)
        AR[1,1] := 'DANG'  ; AR[1,2] := 12
        AR[2,1] := 'BOY'   ; AR[2,2] := 24
        AR[3,1] := 'JOJO'  ; AR[3,2] := 11
        AR[4,1] := 'ANON'  ; AR[4,2] := 18
        ASORT(AR)                   // จดัเรยีงตามชื่อ จากนอยไปมาก
        ASORT(AR,,,{|X,Y|X[2]<Y[2]}) // จดัเรยีงตามอายุจากนอยไปมาก
        AEVAL(AR,{|ELE|QOUT(ELE[1],ELE[2])})
        ASORT(AR,,,{|X,Y|X[2]>Y[2]}) // จดัเรยีงตามอายุจากมากไปนอย
        AEVAL(AR,{|ELE|QOUT(ELE[1],ELE[2])}) // น ําอาเรยมาพิมพ

&  3.20 AT()
! รูปแบบ
AT(<ตวัอกัษรทีต่องการหา>, <ขอความเปาหมาย>) --> ตํ าแหนงที่พบ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงตํ าแหนงของขอความยอย ในขอความที่ตองการ
: ตัวอยางที่ 3.30
        ? AT('A','IJUSTCALLTOSAYILOVEYOU')      // 7
        TXT := 'MR.SILICON VLSI'
        ? SUBSTR(TXT,AT('.',TXT)+1)             // SILICON VLSI
        ? SUBSTR(TXT,1,AT(' ',TXT)-1)           // MR.SILICON
        X := SUBSTR(TXT,1,AT(' ',TXT)-1)
        ? SUBSTR(X,AT('.',X)+1)                 // SILICON

&  3.21 ATAIL()
! รูปแบบ
ATAIL(<ชดุอาเรย>) --> คาของสมาชิกตัวสุดทาย
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาของสมาชิกอาเรยตัวสุดทาย
: ตัวอยางที่ 3.31
        AR := {'C','A','F','D','B'}
        ? ATAIL(AR)                     // B
        ASORT(AR)
        ? 'MAX : ',ATAIL(AR)         // F
        ? 'MIN : ',AR[1]                // A

&  3.22 BOF()
! รูปแบบ
BOF() --> ผลการตรวจสอบเรคอรดแรกของแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบคาเปนจริงเมื่อถึง RECORD แรกของแฟม
: ตัวอยางที่ 3.32
        USE FILEA
        ? BOF() , RECNO()       // .F.       1
        SKIP -1
        ? BOF() , RECNO()       // .T.       1

        GO TOP
        ? BOF() , RECNO()       // .F.       1
        GO BOTTOM
        DO WHILE .NOT. BOF()      //        104       800       700
           ? FIELD1,FIELD2,FIELD3 //        103       785       200
           SKIP -1                //        102      6500       420
        ENDDO                     //        101      1000       150

&  3.23 BREAK()
! รูปแบบ
BREAK() --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ หยุดการทํ างาน
: ตัวอยางที่ 3.33
        USE FILEA
        DO WHILE .NOT. EOF()
           BEGIN SEQUENCE
              ? FIELD1            //        101    ONE RECORD ONLY
              BREAK // TO END OR RECOVER
              SKIP
           END
           BREAK // TO ENDDO OF DO WHILE
        ENDDO
: ตัวอยางที่ 3.34
        USE FILEA
        DO WHILE .NOT. EOF()
           ?? '%'                         // %
           BEGIN SEQUENCE                 // 101       *%
              IF FIELD1 = '103'           // 102       *%
                 BREAK       // WOW*%
              ENDIF      // 104       *
              ? FIELD1
              SKIP      // WHEN IT SKIP, IT WILL GOTO END
           RECOVER      // IT COME FROM BREAK SO IT LOOK LIKE ELSE

              SKIP
              ? 'WOW'   // IT WILL SKIP THIS LINE
           END
           ?? '*'
        ENDDO

&  3.24 BROWSE()
! รูปแบบ
BROWSE([<มมุบน>], [<มมุซาย>], [<มุมลาง>], [<มุมขวา>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงขอมูลแบบ BROWSE ในหนาตางที่กํ าหนด
: ตัวอยางที่ 3.35
        USE FILEA
        BROWSE(5,5,20,70)       // การเรียกดูขอมูลจากแฟม FILEA
        SET FILTER TO FIELD2 >= 800
        BROWSE(6,5,15,70)       // สามารถใช FILTER รวมกับ BROWSE ได

&  3.25 CDOW()
! รูปแบบ
CDOW(<นพิจนวันที>่) --> ชื่อวันในสัปดาห
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที ่เปนวันในสัปดาห
: ตัวอยางที่ 3.36
        ? DATE()                // 12/15/96
        ? DOW(DATE())           //  2
        ? CDOW(DATE())          // MONDAY
        ? CDOW(DATE()-3)        // SATURDAY
        ? DOW(DATE()-3)         //  7
        ? DOW(CTOD("06/04/69")) //  4
        X = "06/30/97"
        ? CDOW(CTOD(X))         // MONDAY

&  3.26 CHR()
! รูปแบบ
CHR(<จ ํานวนเลข>) --> ตัวอักษรในตารางแอสกี (ASCII)
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ แปลงรหัส ASCII เปนตัวอักษร
: ตัวอยางที่ 3.37
        ? CHR(65)              // A
        ? CHR(7)                // SOUND BEEP
        ? CHR(65) + CHR(66)     // AB
        ? CHR(ASC("A")+25)      // Z
        ? "*"                   // *ABCDEFGHIJKLMNOPQRSTUVWXYZ
        FOR I = 0 TO 25
            ?? CHR(ASC("A")+I)
        NEXT

&  3.27 CMONTH()
! รูปแบบ
CMONTH(<นพิจนวนัที>่) --> ชื่อเดือนภาษาอังกฤษ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนชื่อเดือน
: ตัวอยางที่ 3.38
        ? DATE()                                // 03/09/96
        ? CMONTH(DATE())+STR(YEAR(DATE()))   // MARCH 1996
        ? CMONTH(DATE()-2)                      // MARCH
        ? CMONTH(DATE()-10)                     // FEBRUARY
        ? CMONTH(DATE()-250)                    // JULY
        ? CMONTH(CTOD("06/04/69"))              // JUNE
        X = "06/30/97"
        ? CMONTH(CTOD(X))                       // JUNE
        ? SUBSTR(CMONTH(DATE()),1,3)        // MAR

&  3.28 COL()
! รูปแบบ
COL() --> ต ําแหนงหลัก

P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาตํ าแหนงหลักปจจุบัน ไปยังตัวแปร
: ตัวอยางที่ 3.39
        ? COL()         //       0 CURRENT COLUMN =      30     20
        _COL = COL()
        _COL = _COL + 10
        @ ROW() , COL()+1 SAY "CURRENT COLUMN = "
        @ ROW() , COL()+2 SAY COL()
        @ ROW() , COL()+2 SAY _COL

&  3.29 CTOD()
! รูปแบบ
CTOD(<นพิจนอักขระ>) --> วันที่
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงตัวอักษรเปนวันที่
: ตัวอยางที่ 3.40
        X = CTOD("12-31-2002")
        Y = CTOD("12/31/2002")
        ? X,Y                           // 12/31/02 12/31/02
        SET CENTURY ON
        ? X,Y                           // 12/31/2002  12/31/2002
        GDATE = DATE()
        @ ROW() , COL() SAY "GET DATE :" GET GDATE ;
                        RANGE Y , DATE()
        READ

&  3.30 CURDIR()
! รูปแบบ
CURDIR([<ชือ่ไดรฟ>]) --> ชื่อไดเร็คทรอรี่
P หนาที่
ฟงกชนันี้จะทํ าหนาที่ สงคา DIRECTORY ปจจุบัน
: ตัวอยางที่ 3.41
        ? CURDIR("C")                   // LANGUAGE\CLIPPER5
        ? CURDIR("C:")                  // LANGUAGE\CLIPPER5
        ? CURDIR("CAT")                 // LANGUAGE\CLIPPER5

        RUN CD\WINDOWS
        ? CURDIR("C")                   // WINDOWS
        RUN DIR
        RUN CD\
        ? CURDIR("C")+"*"               // *
        RUN CD\LANGUAGE\CLIPPER5
        ? CURDIR("A")                   // DATA

&  3.31 DATE()
! รูปแบบ
DATE() --> วนัที่ของเครื่อง
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาวันที่ปจจุบันของเครื่อง
: ตัวอยางที่ 3.42
        ? DATE()                        // 12/31/02
        SET CENTURY ON
        ? DATE()                        // 12/31/2002

&  3.32 DAY()
! รูปแบบ
DAY(<นพิจนวันที>่) --> วันที่ของเดือน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาวันที่ของเดือน
: ตัวอยางที่ 3.43
        ? DATE()                        // 12/31/02
        ? DAY(DATE())                   // 31
        ? DAY(CTOD(""))                 // 0
        ? DAY(DATE())+1                 // 1

&  3.33 DBAPPEND()
! รูปแบบ
DBAPPEND() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งเพิ่มเรคอรดใหม (เหมือน APPEND BLANK)

: ตัวอยางที่ 3.44
        USE FILEA
        IF RECCOUNT() > 20
           DBAPPEND()
        ELSE
           APPEND BLANK
        ENDIF

&  3.34 DBCLEARFIL()
! รูปแบบ
DBCLEARFIL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งยกเลิกเงื่อนไขปจจุบันของฟวเตอร (เหมือน SET FILTER TO)
: ตัวอยางที่ 3.45
        USE FILEA
        SET FILTER TO FIELD2 >= 800
        COUNT TO X
        ? RECCOUNT(),X          //      4       3
        IF RECCOUNT() < 20
           SET FILTER TO
        ELSE
           DBCLEARFIL()
        ENDIF
        COUNT TO X
        ? RECCOUNT(),X          //      4       4

&  3.35 DBCLEARIND()
! รูปแบบ
DBCLEARIND() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งยกเลิกการใชแฟมดรรชนี INDEX (เหมือน SET INDEX TO)
: ตัวอยางที่ 3.46
        USE FILEA
        INDEX ON FIELD1 TO IFIELD1
        INDEX ON FIELD2 TO IFIELD2
        USE FILEA INDEX IFIELD1,IFIELD2

        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        SET ORDER TO 2
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        SET INDEX TO           // ยกเลิกแฟมแบบ INDEX
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        SET ORDER TO 2         // ไมมีผลอีก เพราะ INDEX ถูกยกเลิกไป
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        DBCLEARIND()           // ยกเลิกแฟมแบบ INDEX
        LIST FIELD1,FIELD2,FIELD3

&  3.36 DBCLEARREL()
! รูปแบบ
DBCLEARREL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งยกเลิก RELATIONS ปจจุบัน (เหมือน SET RELATION TO)
: ตัวอยางที่ 3.47
        USE FILEN INDEX IDN      // ID NAME
        USE FILES INDEX IDS NEW  // ID SALARY
        SET RELATION TO ID INTO FILEN
        LIST  ID , NAME , SALARY
        DBCLEARREL()
        LIST  ID , SALARY

&  3.37 DBCLOSEALL()
! รูปแบบ
DBCLOSEALL() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปดพื้นที่ท ํางานทั้งหมด (เหมือน CLOSE ALL)

: ตัวอยางที่ 3.48
        USE FILEA
        USE FILEB NEW
        LIST FIELD1,FIELD2,FIELD3
        SELE 1
        LIST FIELD1,FIELD2,FIELD3
        DBCLOSEALL()

&  3.38 DBCLOSEAREA()
! รูปแบบ
DBCLOSEAREA() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปดพื้นที่ท ํางานปจจุบัน (เหมือน CLOSE หรือ USE)
: ตัวอยางที่ 3.49
        USE FILEA
        USE FILEB NEW
        LIST FIELD1
        DBCLOSEAREA()
        SELE 1
        LIST FIELD1
        DBCLOSEAREA()

&  3.39 DBCOMMIT()
! รูปแบบ
DBCOMMIT() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งใหเขียนแฟมลงสื่อ มีผลอยางมากตอระบบเครือขาย (เหมือน COMMIT)
: ตัวอยางที่ 3.50
        USE FILEA
        APPEND BLANK
        REPLACE FIELD1 WITH '105',FIELD2 WITH 1250,FIELD3 WITH 100
        DBCOMMIT()

&  3.40 DBCOMMITALL()
! รูปแบบ
DBCOMMITALL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งใหเขียนแฟมในทุกพื้นที่ลงสื่อ มีผลอยางมากตอระบบเครือขาย
: ตัวอยางที่ 3.51
        USE FILEA
        APPEND BLANK
        REPLACE FIELD1 WITH '105',FIELD2 WITH 1250,FIELD3 WITH 100
        USE FILEB NEW
        APPEND BLANK
        REPLACE FIELD1 WITH '105',FIELD2 WITH 1250,FIELD3 WITH 100
        DBCOMMITALL()

&  3.41 DBCREATE()
! รูปแบบ
DBCREATE(<ชือ่แฟมใหม>, <ชุดอาเรยเก็บโครงสราง>) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งสรางแฟมจากโครงสรางอาเรยที่มีอยู
: ตัวอยางที่ 3.52
        FIELDAR := ARRAY(3,4)   // เตรียมใหแฟมนี้มี 3 ฟลด
        FIELDAR[1] := {'FIELD1','C',3,0}
        FIELDAR[2] := {'FIELD2','C',3,0}    ;        FIELDAR[3] := {'FIELD3','C',3,0}
        DBCREATE('TESTCRT',FIELDAR)
        USE TESTCRT
        FLDAR := ARRAY(AFIELDS())
        AFIELDS(FLDAR)
        AEVAL(FLDAR,{|X|QOUT(X)})

&  3.42 DBDELETE()
! รูปแบบ
DBDELETE() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งลบเรคอรดปจจุบัน (เหมือน DELETE)

: ตัวอยางที่ 3.53
        USE FILEA
        DBDELETE()
        PACK

&  3.43 DBEDIT()
! รูปแบบ
DBEDIT([<มมุบน>], [<มมุซาย>], [<มุมลาง>], [<มุมขวา>],
       [<ชดุอาเรยเก็บชื่อฟลดที่ตองการแสดง>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งแสดงเรคอรดที่ตองการ
: ตัวอยางที่ 3.54
        USE FILEA
        AR := {'FIELD1','FIELD2'}
        DBEDIT(5,5,20,70,AR)            // แกไขไมได
        BROWSE(4,5,15,70)               // เพิ่มลบ หรือแกไขได

&  3.44 DBF()
! รูปแบบ
DBF() --> สมนาม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงชื่อแฝงของพื้นที่ท ํางานปจจุบัน
: ตัวอยางที่ 3.55
        USE FILEA ALIAS NAMEFA
        ? DBF()                         // NAMEFA
        USE FILEA
        ? DBF()                        // FILEA

&  3.45 DBFILTER()
! รูปแบบ
DBFILTER() --> ขอความแสดงเงื่อนไขของตัวกรอง (FILTER)
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงเงื่อนไขของ FILTER

: ตัวอยางที่ 3.56
        USE FILEA
        FIL := 'FIELD2 >= 800'
        SET FILTER TO &FIL
        ? DBFILTER()                    // FIELD2 > = 800

&  3.46 DBGOBOTTOM()
! รูปแบบ
DBGOBOTTOM() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ไปยังเรคอรดสุดทาย (เหมือน GO BOTTOM)
: ตัวอยางที่ 3.57
        USE FILEA
        DBGOBOTTOM()
        WHILE !BOF()
           ? FIELD1,FIELD2,FIELD3
           SKIP -1
        END

&  3.47 DBGOTO()
! รูปแบบ
DBGOTO(<เลขทีเ่รคอรดที่ตองการไปชี>้) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ไปยังเรคอรดที่ตองการ (เหมือน GO หรือ GOTO)
: ตัวอยางที่ 3.58
        USE FILEA
        FOR I = 1 TO RECCOUNT()
            DBGOTO(I)   ;            ? FIELD1,FIELD2,FIELD3
        NEXT

&  3.48 DBGOTOP()
! รูปแบบ
DBGOTOP() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ไปยังเรคอรดแรก (เหมือน GO TOP)

: ตัวอยางที่ 3.59
        USE FILEA
        DBGOTO(3)
        ? FIELD1   ;        DBGOTOP()
        WHILE (!EOF())
           ? FIELD1,FIELD2,FIELD3
           DBSKIP()
        END

&  3.49 DBRECALL()
! รูปแบบ
DBRECALL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยกเลิกการทํ าเครื่องหมายลบทั้งหมด (เหมือน RECALL)
: ตัวอยางที่ 3.60
        USE FILEA
        DELETE
        SKIP
        DELETE
        DBRECALL()

&  3.50 DBREINDEX()
! รูปแบบ
DBREINDEX() --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปรับปรุงแฟมดรรชนีใหถูกตอง (เหมือน REINDEX)
: ตัวอยางที่ 3.61
        USE FILEA INDEX IFIELD1,IFIELD2
        DBREINDEX()

&  3.51 DBRELATION()
! รูปแบบ
DBRELATION(<เลขที่แฟม>) --> ชื่อฟลดของแฟมที่ระบุ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แสดงชื่อตัวแปรที่ใชเชื่อมแฟมใหสัมพันธกัน

: ตัวอยางที่ 3.62
        USE FILECUS  INDEX ICUS     // ICUS  NAMEC
        USE FILEEMP  INDEX IEMP NEW // IEMP  NAMEE
        USE FILESALE INDEX ISAL NEW // ISAL  ICUS IEMP PRODUCT
        SET RELATION ICUS INTO FILECUS,IEMP INTO FILEEMP
        ? DBRELATION(1)         // ICUS
        ? DBRELATION(2)         // IEMP

&  3.52 DBRSELECT()
! รูปแบบ
DBRSELECT(<เลขที่แฟม>) --> เลขที่พื้นที่ท ํางาน
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาตัวเลขของพื้นที่ ที่มีการใชฟลดไปเชื่อมความสัมพันธมา
: ตัวอยางที่ 3.63
        USE FILECUS  INDEX ICUS     // ICUS  NAMEC
        USE FILEEMP  INDEX IEMP NEW // IEMP  NAMEE
        USE FILESALE INDEX ISAL NEW // ISAL  ICUS IEMP PRODUCT
        SET RELATION ICUS INTO FILECUS,IEMP INTO FILEEMP
        ? DBRSELECT(1) ,DBRELATION(1)     // 1     ICUS
        ? DBRSELECT(2) ,DBRELATION(2)     // 2     IEMP
        ? ALIAS(DBRSELECT(1))                 // FILECUS
        ? ALIAS(DBRSELECT(2))                 // FILEEMP

&  3.53 DBSEEK()
! รูปแบบ
DBSEEK(<ขอความที่ตองการคนหา>) --> จริงหรือเท็จ
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ คนหาขอมูลตามฟลดที่ถูกจัดเรียงในแฟมดรรชนี
: ตัวอยางที่ 3.64
        USE FILEA INDEX IFIELD1
        ACCEPT TO INDF1
        IF DBSEEK(INDF1)
           ? 'FOUND' ,FIELD1,FIELD2,FIELD3
        ELSE

           ? 'NOT FOUND'
        ENDIF

&  3.54 DBSELECTAR()
! รูปแบบ
DBSELECTAR(<ชือ่พืน้ที>่ | <สมนาม>) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ เลือกพื้นที่เปาหมายตามตองการ
: ตัวอยางที่ 3.65
        USE FILEA
        USE FILEB NEW
        DBSELECTAR("FILEA")
        ? ALIAS() , SELECT()     // FILEA       1

&  3.55 DBSETFILTER()
! รูปแบบ
DBSETFILTER(<เงือ่นไขการสรางตัวกรอง>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ระบุเงื่อนไขในการเลือกขอมูล (เหมือน SET FILTER TO)
: ตัวอยางที่ 3.66
        USE FILEA
        DBSETFILTER({||FIELD2>=800})
        ? DBFILTER()            // ไมมีคาใหแสดง
        LIST FIELD1,FIELD2,FIELD3
        INKEY(0)
        DBSETFILTER({||FIELD2>=800},"FIELD2>=800")
        ? DBFILTER()            // FIELD2>=800
        LIST FIELD1,FIELD2,FIELD3

&  3.56 DBSETINDEX()
! รูปแบบ
DBSETINDEX(<ชือ่แฟมดรรชนี>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งเปดแฟมดรรชนี เพื่อใหขอมูลจัดเรียงตามที่ตองการ (เหมือน SET INDEX TO)

: ตัวอยางที่ 3.67
        USE FILEA
        LIST FIELD1,FIELD2,FIELD3   // แสดงขอมูลโดยไมใชแฟมดรรชนี
        DBSETINDEX("IFIELD2")
        LIST FIELD1,FIELD2,FIELD3    // จดัเรียงตามแฟมดรรชนี IFIELD2

&  3.57 DBSETORDER()
! รูปแบบ
DBSETORDER(<เลขล ําดับของแฟมดรรชนีที่เลือก>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ เลือกแฟมดรรชนีที่เคยเปดไวหลายแฟม (เหมือน SET ORDER TO)
: ตัวอยางที่ 3.68
        USE FILEA INDEX IFIELD2,IFIELD3
        LIST FIELD1,FIELD2,FIELD3    // จดัเรียงตามแฟมดรรชนี IFIELD2
        DBSETORDER(2)
        LIST FIELD1,FIELD2,FIELD3    // จดัเรียงตามแฟมดรรชนี IFIELD3

&  3.58 DBSETRELAT()
! รูปแบบ
DBSETRELAT(<เลขทีพ่ืน้ที>่ | <สมนาม>, <เงื่อนไขการสรางความสัมพันธ>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สั่งเชื่อมพื้นที่ทํ างาน 2 พื้นที่ใหสัมพันธกัน (เหมือน SET RELATION TO)
: ตัวอยางที่ 3.69
        USE FILECUS INDEX ICUS
        USE FILEEMP INDEX IEMP NEW
        DBSETRELAT("FILECUS",{||FILEEMP->IEMP},"FILEEMP->IEMP")
        LIST  IEMP , ICUS , INAMEC , INAMEE

&  3.59 DBSKIP()
! รูปแบบ
DBSKIP([<จ ํานวนเรคอรด>]) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยายตัวชี้เรคอรดไปยังเรคอรดที่ตองการ (เหมือน SKIP)

: ตัวอยางที่ 3.70
        USE FILEA
        DBSKIP()
        DISPLAY FIELD1,FIELD2,FIELD3    // 2  102   6500   420
        DBSKIP(2)
        DISPLAY FIELD1,FIELD2,FIELD3    // 4  104    800   700
        DBSKIP(-1)
        DISPLAY FIELD1,FIELD2,FIELD3    // 3  103    785   200
        DBSKIP(-1)
        DISPLAY FIELD1,FIELD2,FIELD3    // 2  102   6500   420
        DBSKIP(-1)
        DISPLAY FIELD1,FIELD2,FIELD3    // 1  101   1000   150

&  3.60 DBSTRUCT()
! รูปแบบ
DBSTRUCT() --> ชดุอาเรยเก็บโครงสรางแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สราง ARRAY ที่เก็บโครงสรางฐานขอมูล
: ตัวอยางที่ 3.71
        USE FILEA
        STRUC := DBSTRUCT()
        AEVAL(STRUC,{|ELE| QOUT(ELE[1],ELE[2],ELE[3],ELE[4])})
        INKEY(0)
        AEVAL(STRUC,{|ELE| QOUT(ELE[1]) ,QQOUT(ELE[2]),;
                           QQOUT(ELE[3]) ,QQOUT(ELE[4])})

&  3.61 DBUNLOCK()
! รูปแบบ
DBUNLOCK() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยกเลิกการจองพื้นที่ทํ างานปจจุบัน (เหมือน UNLOCK)
: ตัวอยางที่ 3.72
        USE FILEA SHARED
        ? FLOCK()               // .T.

        ? FIELD1                // 101
        DBUNLOCK()              // DBUNLOCK() ตองใชในพื้นที่ที ่LOCK
        USE FILEA SHARED NEW
        ? FLOCK()               // .T.
        SELE 1
        ? FLOCK()               // .F.

&  3.62 DBUNLOCKALL()
! รูปแบบ
DBUNLOCKALL() --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ยกเลิกการจองพื้นที่ทํ างานในทุกพื้นที่ (เหมือน UNLOCK ALL)
: ตัวอยางที่ 3.73
        USE FILEA SHARED
          ? FLOCK()             // .T.
        USE FILEB SHARED NEW
          ? FLOCK()             // .T.
          DBUNLOCKALL()
        USE FILEA SHARED NEW
          ? FLOCK()             // .T.
        USE FILEB SHARED NEW
          ? FLOCK()             // .T.
        SELE 1
          ? FLOCK()             // .F.
        SELE 2
          ? FLOCK()             // .F.

&  3.63 DELETED()
! รูปแบบ
DELETED() --> ผลการตรวจสอบสถานะของเรคอรด
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ บอกใหทราบวาเรคอรดปจจุบัน ถูกทํ าเครื่องหมายใหถูกลบหรือไม

: ตัวอยางที่ 3.74
        USE FILEA
        ? DELETE()              // .F.
        DELETE
        ? DELETE()              // .T.
        DBRECALL()
        ? DELETE()             // .F.

&  3.64 DESCEND()
! รูปแบบ
DESCEND(<นพิจน>) --> คานิพจนเดิม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งเรียงจากมากไปนอยในแฟมดรรชนี
: ตัวอยางที่ 3.75
        USE FILEA
        INDEX ON DESCEND(FIELD3) TO IFIELD3
        LIST FIELD1,FIELD2,FIELD3
        SEEK DESCEND(200)
        IF FOUND()
           DISPLAY FIELD1,FIELD2,FIELD3 // 3   103   785  200
        ENDIF

&  3.65 DEVOUT()
! รูปแบบ
DEVOUT(<นพิจน>, [<ขอความระบุสี>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาออกทางอุปกรณปจจุบัน
: ตัวอยางที่ 3.76
        USE FILEA
        WHILE !EOF()
            DEVPOS(ROW()+1,10)
            DEVOUT(FIELD1+STR(FIELD2)+STR(FIELD3),"BR+/B")
            DBSKIP()
        END

&  3.66 DEVPOS()
! รูปแบบ
DEVPOS(<แถวที>่, <หลักที>่) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ยายตํ าแหนงจากจุดหนึ่งไปอีกจุดหนึ่ง
: ตัวอยางที่ 3.77
        USE FILEA
        I = 1
        WHILE !EOF()
            DEVPOS(I,10)
            DEVOUT(STR(I)+FIELD1+STR(FIELD2)+STR(FIELD3),"GR+/B")
            DBSKIP()
            I++
        END
: ตัวอยางที่ 3.78
        USE FILEA
        SET DEVICE TO PRINTER  // ไมเห็นผลใด ๆ ทางจอภาพ
        WHILE !EOF()
            DEVPOS(PROW()+1,10)
            DEVOUT(FIELD1+STR(FIELD2)+STR(FIELD3),"W+")
            DBSKIP()
        END
        EJECT
        SET DEVICE TO SCREEN

&  3.67 DIRECTORY()
! รูปแบบ
DIRECTORY(<กลุมแฟม>, [<ลักษณะเฉพาะ>]) --> ชุดอาเรยเก็บลักษณะแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สรางอาเรยเก็บชื่อแฟม วันที่ เวลา หรือขนาด
ตวัเลอืกของลักษณะเฉพาะ (ATTRIBUTES)
H  ชือ่แฟม รวมทั้งที่เปนแฟมซอน
S  ชือ่แฟม รวมทั้งที่เปนแฟมระบบ

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


108                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

D  ชือ่แฟม และชื่อ DIRECTORY ทั้งหมด
V  ชือ่อปุกรณเก็บขอมูล ซึ่งเปนชื่อ VOLUMN
ขอมูลที่ไดมา จะแบงเปน 5 สวนคือ
ชือ่แฟม ขนาด วันที่ เวลา ประเภทแฟม
: ตัวอยางที่ 3.79
        OFATTRD := DIRECTORY("C:\*.*","D")
        AEVAL(OFATTRD,{|E|QOUT  (E[1],E[2],E[3],E[4],E[5])})
        ? '---------------------------'

&  3.68 DISKSPACE()
! รูปแบบ
DISKSPACE([<เลขไดรฟ>]) --> ขนาดไดรฟ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบเนื้อที่ที่เหลือในสื่อเก็บขอมูล
: ตัวอยางที่ 3.80
        ? DISKSPACE()           // 81010122  CURRENT DRIVE
        ? DISKSPACE(3)          // 81010122  DRIVE C
        ? DISKSPACE(1)          //  1153555  DRIVE A
        ? DISKSPACE(2)          //  1153555  DRIVE B ถาม ีDRIVE เดียว

&  3.69 DISPBOX()
! รูปแบบ
DISPBOX(<มมุบน>, <มุมซาย>, <มุมลาง>, <มุมขวา>,
        [<แบบของเสน>], [<สีขอบ>]) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงกรอบสี่เหลี่ยมบนจอภาพ
ตวัเลขระบุประเภทกรอบ
1 เสนเดี่ยว
2 เสนคู
: ตัวอยางที่ 3.81
        DISPBOX(5,10,10,30,1,"BG+/B")
        DISPBOX(15,5,20,40,2,"BG+/B")

&  3.70 DISPOUT()
! รูปแบบ
DISPOUT(<นพิจน>, [<สีของตัวอักษร>]) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แสดงสิ่งที่ตองการทางจอภาพ โดยคํ าสั่ง SET DEVICE จะไมมีผลกับค ําสั่งนี้
: ตัวอยางที่ 3.82
        USE FILEA
        I = 1
        WHILE !EOF()
            SETPOS(I,10)
            DISPOUT(STR(I)+FIELD1+STR(FIELD2)+STR(FIELD3),"GR+/B")
            DBSKIP()   ;          I++
        END

&  3.71 DOW()
! รูปแบบ
DOW(<นพิจนวันที>่) --> วันของสัปดาห
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนวันของสัปดาห
: ตัวอยางที่ 3.83
        ? DATE()                // 12/15/96
        ? DOW(DATE())           //  2  (MONDAY)
        ? DOW(DATE()-3)         //  6
        ? DOW(CTOD("06/04/69")) //  4
        FOR I = DOW(DATE()) TO DOW(DATE())+2
            ? CDOW(DATE()+I)    //  TUESDAY WEDNESDAY THURSDAY
        NEXT

&  3.72 DTOC()
! รูปแบบ
DTOC(<นพิจนวันที>่) --> วันที่เปนตัวอักษร
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนตัวอักษร  DD/MM/YY

: ตัวอยางที่ 3.84
        ? DTOC(DATE())                  // 03/05/96
        X = DTOC(DATE())                // 03/01/96
        Y = VAL(SUBSTR(X,4,2))          // 03/02/96
        FOR I = 1 TO Y                  // 03/03/96
            ? DATE()-Y+I                // 03/04/96
        NEXT                            // 03/05/96

&  3.73 DTOS()
! รูปแบบ
DTOS(<นพิจนวันที>่) --> วันที่เปนตัวอักษร
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แปลงวันที่เปนตัวอักษรในรูปแบบ YYYYMMDD
: ตัวอยางที่ 3.85
        ? DTOS(DATE())                  // 19960309
        X = DTOS(DATE())
        ? LEN(X)                        //          8
        ? "YEAR  :"+(SUBSTR(X,1,4))     // YEAR  :1996
        ? "MONTH :"+(SUBSTR(X,5,2))     // MONTH :03
        ? "DAY   :"+(SUBSTR(X,7,2))     // DAY   :09

&  3.74 EMPTY()
! รูปแบบ
EMPTY(<นพิจน>) --> ผลการตรวจสอบคาของนิพจน
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบคาของตัวแปรวามีคาหรือไม
เงือ่นไขการตรวจสอบส ําหรับประเภทของตัวแปร
ARRAY     : กวางเปน 0
CHARACTER : ""
NUMERIC   : 0
DATE      : NULL(CTOD(""))
LOGICAL   : FALSE (.F.)
MEMO      : ""
NIL       : NIL

: ตัวอยางที่ 3.86
        A := {}
        B := {""}
        C := " "
        D := ""
        E := 0
        F := .F.
        ? EMPTY(A)      //  .T.
        IF EMPTY(A)
           ? EMPTY(B)   //  .F.
           ? EMPTY(C)   //  .T.
           ? EMPTY(D)   //  .T.
           ? EMPTY(E)   //  .T.
           ? EMPTY(F)   //  .T.
        ENDIF

&  3.75 EOF()
! รูปแบบ
EOF() --> ผลการตรวจสอบเรคอรดสุดทายของแฟม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ถาหมดเรคอรดสุดทายจะสงคาจริงมาให
: ตัวอยางที่ 3.87
        USE FILEA
        ? EOF() , RECNO()       // .F.       1
        GO BOTTOM
        ? EOF() , RECNO()       // .F.       4
        SKIP
        ? EOF() , RECNO()       // .T.       5
        GO TOP
        DO WHILE .NOT. EOF()      //        101      1000       150
           ? FIELD1,FIELD2,FIELD3 //        102      6500       420
           SKIP                   //        103       785       200
        ENDDO                     //        104       800       700

&  3.76 ERRORLEVEL()
! รูปแบบ
ERRORLEVEL([<คาผดิพลาด>]) --> คาผิดพลาด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ระบุคาที่จะสงกลับใหโปรแกรม
: ตัวอยางที่ 3.88
        IF !FILE("XX.PRG")
           ERRORLEVEL(25)       // TO SET VALUE OF ERRORLEVEL
           ERR = ERRORLEVEL()
        ENDIF
        IF ERRORLEVEL() = 25
           RUN DIR/S/OD/W
        ENDIF
        ? ERR                   //     25

&  3.77 EVAL()
! รูปแบบ
EVAL(<นพิจนค ํานวณ>, [<รายการคาตาง ๆ ที่ปอนใหนิพจน>]) --> คาที่ไดจากนิพจน
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ ตรวจสอบโปรแกรม
: ตัวอยางที่ 3.89
        BCODE = {|A|A+1}
        ? EVAL(BCODE,5)         //     6
        ? EVAL(BCODE,2)         //     3
        ? EVAL(BCODE,7)         //     8
        ? (BCODE,5)             //     5
        ? BCODE                 //    NOTHING
: ตัวอยางที่ 3.90
        BCODE = &("{|A|A*5-2}")
        FOR I = 1 TO 3          //     3
           ? EVAL(BCODE,I)      //     8
        NEXT                    //    13

: ตัวอยางที่ 3.91
        GRADE = {|G|IF(G>80,'A',(IF(G>70,'B','F')))}
        ? EVAL(GRADE,85)        // A
        ? EVAL(GRADE,75)        // B
        ? EVAL(GRADE,65)        // F

&  3.78 EXP()
! รูปแบบ
EXP(<นพิจนตวัเลข>) --> ผลการยกกํ าลังของคา E
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งคํ านวณ E ยกกํ าลัง X
: ตัวอยางที่ 3.92
        ? EXP(1)                //      2.72
        SET DECIMAL TO 10
        ? EXP(1)                //      2.7182818285
        ? LOG(EXP(5))           //      5.0000000000
        ? EXP(LOG(5))           //      5.0000000000

&  3.79 FCLOSE()
! รูปแบบ
FCLOSE(<ชือ่ตวัแปรระดับแฟม>) --> ผลการปดแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สั่งปดแฟมที่เปดแบบ BINARY แลว น ําคาใน BUFFER เขียนลง DISK
: ตัวอยางที่ 3.93
        F := FCREATE("TEST.TXT",0)  // เปดแฟมแบบอานเขียนได
        FWRITE(F,"COMPUTER",10)     // แฟมจะมีขนาด 10 ไบต
        FCLOSE(F)

&  3.80 FCOUNT()
! รูปแบบ
FCOUNT() --> จ ํานวนฟลดของแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ จะแสดงจํ านวณฟลดของแฟม

: ตัวอยางที่ 3.94
        USE FILEA
        AFLDNAME := ARRAY(AFIELDS())
        AFLDTYPE := ARRAY(AFIELDS())
        AFLDWIDTH:= ARRAY(AFIELDS())
        AFLDDEC  := ARRAY(AFIELDS())
        AFIELDS(AFLDNAME,AFLDTYPE,AFLDWIDTH,AFLDDEC)
        FOR I = 1 TO AFIELDS()                  // FIELD1 C  10
            ?  AFLDNAME[I],AFLDTYPE[I]          // FIELD2 N  10  0
            ?? STR(AFLDWIDTH[I],10)             // FIELD3 N  10  0
            ?? IF(AFLDTYPE[I]='N',STR(AFLDDEC[I],10),'')
        NEXT
        ? AFIELDS()     //     3
        ? FCOUNT()      //     3

&  3.81  FCREATE()
! รูปแบบ
FCREATE(<ชือ่แฟม>, [<เลขระบุลักษณะเฉพาะ>]) --> ตัวแปรระดับแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สรางแฟมแบบ BINARY FILE
การสรางแฟมสามารถระบุแฟมไดหลายแบบดังนี้
0  อาน/เขียนได
1  อานอยางเดียว
2  แฟมแบบซอน
3  แฟมระบบ
: ตัวอยางที่ 3.95
        F := FCREATE("TEST.TXT",0)  // เปดแฟมแบบอานเขียนได
        FWRITE(F,"AA",10)
        FWRITE(F,"AA",10)
        FWRITE(F,"AA",10)
        FWRITE(F,"AA",10)
        FCLOSE(F)                   // แฟมจะมีขนาด 40 ไบต
        ? FERROR()                  // 0 แสดงวาปดแฟมไดไมผิดพลาด

&  3.82 FERASE()
! รูปแบบ
FERASE(<ชือ่แฟม>) --> ผลการลบแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งลบแฟมจากสื่อเก็บขอมูล
: ตัวอยางที่ 3.96
        IF FERASE("TEST.TXT") = -1
           ? "ERROR ON ERASING"
        ENDIF

&  3.83 FERROR()
! รูปแบบ
FERROR() --> เลขระบุความผิดพลาด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบขอผิดพลาดของการทํ างานกับ BINARY FILE
: ตัวอยางที่ 3.97
        F := FCREATE("TEST.TXT",0)  // เปดแฟมแบบอานเขียนได
        IF FERROR() = 0
           FWRITE(F,"AA",10)
           FWRITE(F,"AA",10)
           FWRITE(F,"AA",10)
           FWRITE(F,"AA",10)
           FCLOSE(F)                // แฟมจะมีขนาด 40 ไบต
           ? FERROR()               // 0 แสดงวาปดแฟมไดไมผิดพลาด
        ENDIF

&  3.84 FIELDNAME()
! รูปแบบ
FIELDNAME(<ต ําแหนงฟลด>) --> ชื่อฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงชื่อฟลดจากแฟมปจจุบัน
: ตัวอยางที่ 3.98
        USE FILEA
        ? FIELDNAME(1)          // FIELD1 แสดงชื่อฟลดแรกของแฟม
        FOR I = 1 TO FCOUNT()

            ? FIELDNAME(I)
        NEXT

&  3.85 FIELDGET()
! รูปแบบ
FIELDGET(<ต ําแหนงฟลด>) --> คาของฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ รับคาจากจากฟลด
: ตัวอยางที่ 3.99
        USE FILEA
        ? FIELD1                // 101
        ? FIELD2                //       1000
        ? FIELDGET(1)           // 101
        ? FIELDGET(2)           //       1000
        X := FIELDNAME(1)
        Y := &X
        ? Y                     // 101

&  3.86 FIELDPOS()
! รูปแบบ
FIELDPOS(<ชือ่ฟลด>) --> ตํ าแหนงฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงตํ าแหนงของฟลดปจจุบัน
: ตัวอยางที่ 3.100
        USE FILEA
        ? FIELDPOS("FIELD1")             //        1
        ? FIELDPOS("FIELD2")             //        2
        ? FIELDGET(FIELDPOS("FIELD1")) // 101

&  3.87 FIELDPUT()
! รูปแบบ
FIELDPUT(<ต ําแหนงฟลด>, <คาใหม>) --> คาใหมที่แทนที่คาเกาในฟลด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ปรับปรุงคาในฟลด

: ตัวอยางที่ 3.101
        USE FILEA
        ? FIELDGET(1)       // 101
        FIELDPUT(1,"103")   // ท ําใหคาในฟลดเปลี่ยนไปโดยไมตอง REPLACE
        ? FIELDGET(1)       // 103

&  3.88 FILE()
! รูปแบบ
FILE(<ชือ่แฟม>) --> ผลการตรวจสอบการเปดแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบการเปดแฟม
: ตัวอยางที่ 3.102
        ? FILE("COMMAND.COM")           // .F.
        ? FILE("\WINDOWS\WIN.COM")      // .T.
        SET PATH TO \WINDOWS
        ? FILE("SYSTEM.INI")            // .T.
        SET DEFAULT TO C:\DOS
        ? FILE("COMMAND.COM")           // .T.
        SET PATH TO \WINDOWS
        SET DEFAULT TO C:\DOS
        ? FILE("XCOPY.EXE")             // .T.
        ? FILE("WIN.INI")               // .T.
        ? CURDIR("C:")                  // LANGUAGE\CLIPPER5

